
Every so often as you flick through the AS400 and iSeries trade magazines, or browse through 
the FAQ's and bulletin boards on the internet you will come across a tip reminding you that it is 
possible to modify the layout of the AS400 or iSeries sign-on display. 
 
My usual recommendation is to "remove" the, in my opinion, superfluous input fields for the 
current library, menu and program/procedure and to use the space for presenting system news 
and announcements to our users as can be seen from the screen shot below.  In addition I also 
recommend adding a support line number for your users and a "legal" warning to discourage 
unauthorised access to your system. 
 

 
Figure 1: Example sign-on screen display with changes implemented 

In this series of articles I will be showing you how to create an automated system news and 
announcement utility that will help to keep your users aware of important system events and 
really get the message across. 
 

Changing the Sign-On Display 
The first thing that we need to do is to customise the sign-on display so that it fits our needs.  
The source for the sign-on display can be found in source file QDDSSRC in library QGPL and is 
stored in source member QDSIGNON. 
 
The first thing to do is to copy the source member QDSIGNON to another location for editing and 
for use.  In our shop we have a library that sits above QSYS in the system library list, and this 
holds all system wide utilities, replacement commands, etc.  Over the following pages, you will 
need to replace each occurrence of "library_name" with the name of the library you use at your 
installation. 
 
To copy the source file from QGPL to your own library, use the CPYF (Copy File) command: 
 
CPYF FROMFILE(QGPL/QDDSSRC) TOFILE(library_name/QDDSSRC) FROMMBR(QDSIGNON) 
 MBROPT(*ADD) 
 
I'm assuming here that you don't already have a version of QDSIGNON in the library you're 
copying to.  If you do, the MBROPT(*ADD) parameter will generate an exception error and you 
should determine whether to change the existing member or not.  You may want to specify a 
new member name and should add the TOMBR parameter to the above statement as follows: 
 
CPYF FROMFILE(QGPL/QDDSSRC) TOFILE(library_name/QDDSSRC) FROMMBR(QDSIGNON)   

TOMBR (new_member) MBROPT(*ADD) 
 
Now you have a copy of the source, it's time to make some changes.  Figure 2, below, shows 
the modified version of source member QDSIGNON.  The important thing to remember when 
modifying QDSIGNON is that under no circumstances should you remove any existing 
fields from the source.  In other words, even though we don't want to have access to the 
PROGRAM, MENU and CURLIB fields we should not comment them out or delete the 
statements from the source member.  Instead you have to relocate them and set their Non-
Display and Protected display attributes. 
 
So, first, change the "Sign-On" title on line 1 of the display to your company name or some other 
identifying text.  It's not strictly necessary, but can be especially useful where you have multiple 
servers performing different functions. 
 
     A                                      DSPSIZ(24 80 *DS3) 
     A* 
     A          R SIGNON 
     A                                      CLEAR 
     A                                      BLINK 
     A                                  1 30'Astradyne (UK) Limited' 
     A                                      DSPATR(HI) 
     A                                  2 48'System  . . . . . :' 
     A            SYSNAME        8A  O  2 70 
     A                                  3 48'Subsystem . . . . :' 
     A            SBSNAME       10A  O  3 70 
     A                                  4 48'Display . . . . . :' 
     A            DEVNAME       10A  O  4 70 
     A                                  6 17'User  . . . . . . . . . . . . . .' 
     A            USERID        10A  B  6 53 
     A  01                              7 17'Password  . . . . . . . . . . . .' 
     A  01        PASSWRD       10A  I  7 53DSPATR(ND) 
     A****                              8 17'Program/procedure . . . . . . . .' 



     A            PROGRAM       10A  B  8 18CHECK(LC) 
     A                                      DSPATR(ND) 
     A                                      DSPATR(PR) 
     A****                              9 17'Menu  . . . . . . . . . . . . . .' 
     A            MENU          10A  B  8 38CHECK(LC) 
     A                                      DSPATR(PR) 
     A****                             10 17'Current library . . . . . . . . .' 
     A                                      DSPATR(ND) 
     A            CURLIB        10A  B  8 58CHECK(LC) 
     A                                      DSPATR(PR) 
     A                                      DSPATR(ND) 
     A  02        QSNERROR      80A  O 24  1DSPATR(HI) 
     A            COPYRIGHT     40A  O 24 40DSPATR(HI) 
     A            UBUFFER      128A  H 
     A            HDESK1        32A  O  2  8MSGID(SGN 0016 *LIBL/AUKMSGF) 
     A            HDESK2        32A  O  3  2MSGID(SGN 0017 *LIBL/AUKMSGF) 
     A            HDESK3        32A  O  4  2MSGID(SGN 0018 *LIBL/AUKMSGF) 
     A            SGN000        78A  O  9  2MSGID(SGN 0000 *LIBL/AUKMSGF) 
     A                                      DSPATR(HI) 
     A            SGN001        78A  O 10  2MSGID(SGN 0001 *LIBL/AUKMSGF) 
     A                                      DSPATR(HI) 
     A            SGN002        78A  O 11  2MSGID(SGN 0002 *LIBL/AUKMSGF) 
     A            SGN003        78A  O 12  2MSGID(SGN 0003 *LIBL/AUKMSGF) 
     A            SGN004        78A  O 13  2MSGID(SGN 0004 *LIBL/AUKMSGF) 
     A            SGN005        78A  O 14  2MSGID(SGN 0005 *LIBL/AUKMSGF) 
     A            SGN006        78A  O 15  2MSGID(SGN 0006 *LIBL/AUKMSGF) 
     A            SGN007        78   O 16  2MSGID(SGN 0007 *LIBL/AUKMSGF) 
     A            SGN999        78A  O 17  2MSGID(SGN 9999 *LIBL/AUKMSGF) 
     A                                      DSPATR(HI) 
     A            WARN01        78   O 19  2MSGID(SGN 0010 *LIBL/AUKMSGF) 
     A                                      COLOR(RED) 
     A            WARN02        78   O 20  2MSGID(SGN 0011 *LIBL/AUKMSGF) 
     A            WARN03        78   O 21  2MSGID(SGN 0012 *LIBL/AUKMSGF) 

Figure 2: QDSIGNON Source Member 

 
The second change to make is to remove the label text for the PROGRAM, MENU and CURLIB 
fields so that we can reclaim the space.  As these are constants they wont affect the I/O buffer 
so it is safe to remove the statements from the source member.  Next, add DSPATR(ND) and 
DSPATR(PR) attributes to each of the PROGRAM, MENU and CURLIB fields, and move them 
all to line 8 of the display starting in positions 18, 38 amd 58 respectively. 
 
The third area to change is to add variables for the additional fields/constants that we want to 
output to the sign-on display.  For this we use MSGID's (as opposed to MSGCON's) to load the 
required text into the output fields.  Using MSGCON's would have required that the display file is 
recompiled every time the message text changes.  MSGID avoids this problem. 
 
The lines added in this area define the Help-Desk contact information, an announcement/notice 
area and a warning against unauthorised access.  Once the changes have been made save the 
source, but don't compile just yet - we've other things to do first. 
 

Creating the Message File 
Before compiling the QDSIGNON display file we should create the message file and configure 
the message texts that are going to be used on the sign-on display.  To create the message file, 
use the CRTMSGF (Create Message File) command from any command line as follows: 
 
CRTMSGF MSGF(library_name/AUKMSGF) TEXT('Sign-on Display Announcement Message 

File') 
 



If you change the message file name from AUKMSGF, don't forget to change it within the 
QDSIGNON source member and also when running the following commands to create the 
messages within the message file: 
 
ADDLIBLE LIB(library_name) POS(*FIRST) 
 
ADDMSGD MSGID(SGN0000) MSGF(AUKMSGF) MSG('_________________________________________ 

_____________________________________') 
 

 
Note, there are 78 underline characters in the above MSG parameter. 
 
ADDMSGD MSGID(SGN0001) MSGF(AUKMSGF) MSG(' +++++++ AS/400 SYSTEM NEWS +++ 07 APRIL, 

 2003 +++ AS/400 SYSTEM NEWS +++++++') 
 
ADDMSGD MSGID(SGN0002) MSGF(AUKMSGF) MSG('                               FASCINATIN 

G FACT') 
 
ADDMSGD MSGID(SGN0003) MSGF(AUKMSGF) MSG(' ') 
 
ADDMSGD MSGID(SGN0004) MSGF(AUKMSGF) MSG(' No. 69 - The electric chair was invented 

 by a dentist.') 
 
ADDMSGD MSGID(SGN0005) MSGF(AUKMSGF) MSG(' ') 
 
ADDMSGD MSGID(SGN0006) MSGF(AUKMSGF) MSG(' ') 
 
ADDMSGD MSGID(SGN0001) MSGF(AUKMSGF) MSG(' +++++++ AS/400 SYSTEM NEWS +++ 07 APRIL, 

 2003 +++ AS/400 SYSTEM NEWS +++++++') 
 
ADDMSGD MSGID(SGN0010) MSGF(AUKMSGF) MSG('WARNING') 
 
ADDMSGD MSGID(SGN0011) MSGF(AUKMSGF) MSG('You are attempting to access a private co 

mputer system.  If you have not been') 
 
ADDMSGD MSGID(SGN0012) MSGF(AUKMSGF) MSG('authorised to access this system you shou 

ld not proceed any further.') 
 
ADDMSGD MSGID(SGN0016) MSGF(AUKMSGF) MSG('Help-Desk') 
 
ADDMSGD MSGID(SGN0017) MSGF(AUKMSGF) MSG('Tel: +44-1252-400955') 
 
ADDMSGD MSGID(SGN0018) MSGF(AUKMSGF) MSG('Fax: +44-1252-400956') 
 
ADDMSGD MSGID(SGN0000) MSGF(AUKMSGF) MSG('~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~ 

~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~') 
 
Now that the message file has been created and the latest announcement has been added you 
can compile the sign-on display source member, QDSIGNON. 
 

Passing the Message 
Finally, for this instalment anyway, you're message file has been created, you're version of 
QDSIGNON has been changed and compiled, there's only one more thing left to do - activate 
the new sign-on display. 
 
To activate the new sign-on display, you have to assign it to an interactive subsystem, e.g. 
QINTER, QCTL, etc, and you can do this using the Change Subscription Description 
(CHGSBSD) command: 
 
CHGSBSD SBSD(QINTER) SGNDSPF(library_name/QDSIGNON) 
 



Note, however, that the sign-on display will not become active until the subsystem has been 
ended and re-started. 
 
Before we finish, a word of caution.  Don't change the Sign-on Display File assigned to your 
controlling subsystem (usually QCTL), as the system console will usually sign into the controlling 
subsystem it is a wise precaution to keep at least one terminal available with the option to 
specify a program or procedure at sign-on time.  I have seen situations where a bug has been 
introduced to a system wide initial program and the ability to change the initial program at sign-
on time was needed to allow access to the system. 
 
That's it for this instalment, you can now change your system announcements and notices at will 
simply editing the message file using the WRKMSGF (Work with Message File) command.  In 
the next article we will be creating a utility to edit the message descriptions as an alternative to 
the WRKMSGF command, allowing us to incorporate display attributes and colour in the body of 
the message. 
 
 
 


